
Paradyn Paral le l  Performance Tools

MDL Programmer’s Guide 4/13/04

Paradyn Project
Computer Sciences Department
University of Wisconsin
Madison, WI 53706-1685
paradyn@cs.wisc.edu

Release 4.1
April 2004

MDL
Programmer’s Guide



Table of Contents

MDL Programmer’s Guide April 13, 2004 Release 4.1

1 Introduction......................................................................................................................3
2 Counters ...........................................................................................................................3

2.1 Counters are initialized to zero ...........................................................................4
3 Timers ..............................................................................................................................5

3.1 Every startTimer() must be matched by a stopTimer() .......................................5
3.2 Use append/append instead of append/prepend for timers .................................6

4 Entry-Return Instrumentation ..........................................................................................8
4.1 Return-point execution does NOT imply entry-point execution .........................8
4.2 Entry-point execution does NOT imply return-point execution .........................9
4.3 Avoid using entry-point post-instrumentation ....................................................11
4.4 Do NOT use return-point post-instrumentation ..................................................12

5 Callsite Instrumentation ...................................................................................................13
5.1 In exclusive metrics, preInsn callsite is analogous to func.return .......................13
5.2 In exclusive metrics, postInsn callsite is analogous to func.entry ......................14

6 Miscellaneous ..................................................................................................................14
6.1 All EventCounter-style metric values must increase monotonically ..................14
6.2 Use SampledFunction-style metrics to track values in a user application ..........15
6.3 Instrumenting recursive functions .......................................................................16

7 Conclusion .......................................................................................................................16



Page 3

MDL Programmer’s Guide April 13, 2004 Release 4.1

1 INTRODUCTION

Paradyn is a parallelperformancemeasurementtool that performsdynamicinstrumentationon
running applications. Instrumentationis a sequenceof instructions,suchas a code fragment
insertedat thebeginningof a functionto incrementacountof thenumberof timesthefunctionis
invoked. The instructionsarecompiledfrom a sourcelanguage,called the Metric Description
Language (MDL). For Paradynto properlyinstrument(andthusmeasure)your application’s per-
formance,you mustspecifyall themetricsyou will needusingMDL in a Paradynconfiguration
file. Commonlyusedmetricshavealreadybeenprogrammedfor youaspartof theParadyndistri-
bution in a file namedparadyn.rc.

Thispaperis written to helpyoubecomemoreproficientin programmingwith MDL. Specif-
ically, thispaperis designedto saveyoutimeandfrustrationsin usingthisnew language.MDL is
unlike any languagethatyou might have used,sowe want to easeyour introduction.We assume
you have alreadyfamiliarizedyourselfwith theMDL languagesyntax:if not, pleaserefer to the
Paradyn User’s Guide. We further assumeyou are proficient in programming;MDL is an
unusualprogramminglanguage,asyou will shortly discover, andunderstandingadvancedpro-
gramming concepts will ease your learning.

This paperwalksyou throughfrom thebottomup. We first startwith thebasicMDL metric
datatypes: countersandtimers. The restof the paperessentiallygoesthroughtips on how to
properlyusethem. Even seasonedprogrammerswill be surprisedby how subtlethe program-
ming issuesarewhenusingevenseeminglysimpleconceptssuchascountersandtimers. Since
thecountersandtimersdo not updatethemselvesautomatically, we go on to actualinstrumenta-
tion code that updates them.

Threetypesof instrumentationcanbespecifiedin MDL, with eachtypenamedby their loca-
tion (or point of insertion): entry-point,return-point,andcallsite. Entry-pointinstrumentationis
insertedat thebeginningof functions. Return-pointinstrumentationis insertedat theendof func-
tions. Callsite instrumentationis insertedat placeswherefunctionscall other functionswhen
dealingwith exclusive metrics. Entry-pointandreturn-pointinstrumentationarecloselyrelated,
so we considerthemtogetherin onesection. Callsiteinstrumentationis describedin a separate
section.

Last,we devotea sectionon miscellaneousissuesthatdid not fit nicely into any of theabove
mentioned sections and did not warrant separate sections themselves.

To simplify the presentation,we will be providing only MDL codefragments,andnot com-
pletemetric implementations.In addition,thoughconstraintsarekey to instrumentation,noneof
the MDL examplesareconstrained. That is, all exampleswill be written asif for whole-pro-
grammetrics. Again, this is doneto simplify thetext. Themetricsyou write, however, will usu-
ally need to beconstrained.

2 COUNTERS

Countersareoneof two metricdatatypessupportedby MDL. Thesecond,timers,will becov-
ered in Section3.

2.1 Counters are initialized to zero

Therefore, if you are using a counter as a flag, try NOT to use zero as the “okay” value.

Hereis a fragmentof a metricthatcountsthenumberof I/O functioncalls,exceptthosecalls
madefrom within MPI functions.(MPI is a library of functionsusedfor exchangingvaluesin par-
allel programs.)

  // Good example of using a counter as a flag

  counter not_in_mpi;  // Flag is 1 when program is outside an MPI function

  foreach func in mpi_funcs {
    append  preInsn func.entry  (* not_in_mpi = 0; *)
    prepend preInsn func.return (* not_in_mpi = 1; *)
  }

  foreach func in io_funcs {
    append preInsn func.entry (* if (not_in_mpi == 1) io_ops++; *)
  }

The not_in_mpi flag is setonly after a returnfrom an MPI function,andwe incrementthe
io_ops counteronly if the flag is on. That is, we areincrementingthe counterof I/O function
callsonly whenwearesurewearenotexecutingwithin anMPI function. AssumingthattheMPI
functionsdo not call eachother, directly or indirectly, it is impossiblefor thenot_in_mpi flag to
have a valueof 1 whenwithin an MPI function. To show this, we needa caseanalysis. At the
momentof instrumentation,theprogramis eitherexecutinganMPI functionor it is not, sothere
are only two casesto consider. (Note that if a programis currently within a function at the
momentof instrumentation,any new entry-pointinstrumentationof thefunctionmaynot beexe-
cuted for the current invocation of the function.)

Case 1:  Program is inside an MPI function at the moment of instrumentation.
Sincenot_in_mpi is initialized to zero, the flag has the correct off value.  As a result, we
will not mistakenly increment the I/O function calls count when inside an MPI function.

Case 2:  Program is outside an MPI function at the moment of instrumentation.
Sincenot_in_mpi is initializedto zero,theflaghastheincorrectoff value.However, thismet-
ric was written for an MPI application, so this situation will be corrected shortly at the next
call of an MPI function.
Supposewe mistakenly usedzeroasthe“okay” flag valuefor this metric,asin thefollowing

example.

  // Bad example of using a counter as a flag

  counter in_mpi;  // Flag is 1 when program is inside an MPI function

  foreach func in mpi_funcs {
    append  preInsn func.entry  (* in_mpi = 1; *)
    prepend preInsn func.return (* in_mpi = 0; *)

  }

  foreach func in io_funcs {
    append preInsn func.entry (* if (in_mpi == 0) io_ops++; *)
  }

Using a case analysis, we find that it is possible to mistakenly count I/O function calls made
from within MPI functions.

Case 1:  Program is inside an MPI function at the moment of instrumentation.
Since in_mpi is initialized to zero, the flag has the incorrect off value.  As a result, we
might mistakenly increment the I/O function calls count when inside the currently executing
MPI function.

Case 2:  Program is outside an MPI function at the moment of instrumentation.
Since in_mpi is initialized to zero, the flag has the correct off value.

You may notice that deciding on the “okay” value of a flag depends on the errors the metric
must avoid. In the above example, it was required that I/O function calls by MPI functions not be
counted, so we had to choose 1 as the “okay” value.  However, there may be situations where you
must use 0 as the “okay” value.

3 TIMERS

Timers are more complex than counters, but they still can be used with a few simple rules. Para-
dyn has four timer functions: startProcessTimer() and stopProcessTimer() for virtual clock
timers, and startWallTimer() and stopWallTimer() for wall clock timers. For clarity, we
shall sometimes use the abbreviations startTimer() and stopTimer().

3.1 Every startTimer() must be matched by a stopTimer()

This is similar to the syntax of many programming languages. In Pascal, a BEGIN is ended by the
first unmatched END. In C, an open brace { is ended by the first unmatched close brace }. In Lisp,
an open parenthesis ( is ended by the first unmatched close parenthesis ). In MDL, a start-

Timer(T), for some timer T, is ended by the first unmatched stopTimer(T). An unmatched
stopTimer() has no effect. Calling startTimer() on an already running timer neither restarts
the timer nor starts a new copy of the timer. However, the timer is actually stopped ONLY when
every startTimer() has been matched by a stopTimer().

Unlike the Pascal, C, and Lisp examples, MDL timer matching is done as the instrumentation
is executed, and not when the metric was written by you. That is, unmatched elements in Pascal,
C, and Lisp will be caught by a compiler, but unmatched MDL timers will result in incorrect timer
values. At this point, we just want you to be aware of the above rule. The later sections on instru-
mentation will show you how to keep startTimer() and stopTimer() matched.

3.2 Use append/append instead of append/prepend for timers

Software instrumentationinherently perturbsthe original program, so the program’s timing
behavior changes.Furthermore,subsequentinstrumentationcan changethe timing behavior of
previousinstrumentation.Thefollowing exampleshows how relative errorsin timerscanbemin-
imized by careful relative placementof timer instrumentationblocks. Let us illustrate this by
examining instrumentation that times the execution of an I/O function.

We startwith an I/O function that hasalreadybeeninstrumentedasfollows. E1 andE2 are
two piecesof entry-pointinstrumentation.R1 andR2 aretwo piecesof return-pointinstrumenta-
tion.

Now we want to insert our timer instrumentation, written in MDL as follows.

  // The most obvious way to implement an I/O function timer metric

  foreach func in io_funcs {
    append  preInsn func.entry  (* startWallTimer(T); *)
    prepend preInsn func.return (* stopWallTimer(T);  *)
  }

The following figure shows how thingswould look after we have actually instrumentedthe
I/O function with the code given above.

E1 E2

R1 R2

I/O Func

E1 E2

R1 R2

I/O Func startWallTimer(T)

stopWallTimer(T)



Page 7

MDL Programmer’s Guide April 13, 2004 Release 4.1

Notice that for a single timer metric, this is the bestplacementof the instrumentation.By
appending startWallTimer(T), we avoid including the executiontimes of the existing entry-
point instrumentation. By prepending stopWallTimer(T), we avoid including the execution
timesof the existing return-pointinstrumentation.However, let us now considerwhat happens
whenmultiple timersareinserted.For example,we mayhave insertedonetimer to keeptheI/O
time for theentireprogram,onetimer to keeptheI/O time for amodule(asetof procedures),and
one timer to keep the I/O time for a particular procedure.

Below we have instrumentationfor timersT1, T2, andT3; inserted,in thatorder. For thesake
of clarity, let us assume there was no preexisting instrumentation on the I/O function.

Notice thatT3 getsthebestplacementfor beinginsertedlast. Meanwhile,T1 getstheworst
placementafterbeinginsertedfirst. As new timersareinserted,theearliertimersgetlessandless
accurate because they must also timeall the new timers’ instrumentation.

To reducethe rate of decreasein timer accuracies,we can useappend/append insteadof
append/prepend in our metrics.  That is, the MDL description becomes the following.

  // An I/O function timer metric that causes less interference

  foreach func in io_funcs {
    append preInsn func.entry  (* startWallTimer(T); *)

append preInsn func.return (* stopWallTimer(T);  *)
  }

Hereagain we have instrumentationfor timersT1, T2, andT3; inserted,in that order. How-
ever, this time, the return-pointinstrumentationhasbeenappended, insteadof prepended. As
new timersareinserted,theearliertimersstill get lessaccurate,but therateof decreasein accu-
racy hasbeenlessenedbecausethey musttime only half thenew timers’ instrumentation.Also,
noticethat if all the instrumentationpiecestake the sameamountof time to execute,eachtimer
will seethesameamountof error. This maybebetterbecausethenyou won’t have metricsdis-
agreeing with each other about the amount of time this I/O function takes.

I/O Func startTimer(T1) startTimer(T2) startTimer(T3)

stopTimer(T3) stopTimer(T2) stopTimer(T1)



Page 8

MDL Programmer’s Guide April 13, 2004 Release 4.1

Using prepend/prepend also has the same relative-error minimization property, but reduces
the chances of instrumentation being executed at the earliest possible opportunity. To see this, the
following figure illustrates the situation after instrumentation for T1 and T2 have been inserted
with prepend/prepend. Suppose the program is executing the startTimer(T2) instrumentation
when we need to insert instrumentation for T3. Since startTimer(T3) will be prepended in
front of startTimer(T2), startTimer(T3) will not get executed for this invocation of the I/O
function. However, had we instead been appending, the startTimer(T3) will get executed for
this invocation.

4 ENTRY-RETURN INSTRUMENTATION

In the previous sections, we looked at the basic properties of counters and timers. In this section,
we look at how we can properly update the counters and timers in our entry-point and return-point
instrumentation.  Section 5 will cover callsite instrumentation.

4.1 Return-point execution does NOT imply entry-point execution

The simplest example is the case where the program was executing within the function when the
function becomes instrumented. The entry-point instrumentation will not get executed for this
invocation of the function because the entry point has already been passed. However, the return-

I/O Func startTimer(T1) startTimer(T2) startTimer(T3)

stopTimer(T1) stopTimer(T2) stopTimer(T3)

I/O Func startTimer(T2) startTimer(T1)

stopTimer(T2) stopTimer(T1)

point instrumentation may get executed as part of the function’s return. We use an example to
illustrate how to handle this situation.

Here are the most obvious MDL statements for an inclusive constraint flag. constraintFlag

is nonzero if there is any known active invocation of the instrumented function.

  counter constraintFlag;

  append preInsn func.entry  (* constraintFlag += 1; *)
  append preInsn func.return (* constraintFlag -= 1; *)

Let us make a simple walk-through of the above MDL statements. As with all counters, con-
straintFlag is initialized to zero. On entry to the instrumented function, constraintFlag is
incremented to 1. On return from the instrumented function, constraintFlag is decremented
back to 0. Should the instrumented function be recursive, the recursive invocation will first
increase constraintFlag on entry, but will properly restore constraintFlag on return. It
seems that the above MDL statements correctly give constraintFlag a nonzero value when
there is an active invocation of the function.

However, suppose that the program was executing within the function at the moment of instru-
mentation. Then the entry-point instrumentation will not be executed, but the return-point instru-
mentation is. The result is that constraintFlag gets set to -1 at return. Now constraintFlag is
nonzero even when there is no active invocation of the function. It gets worse. On the next invo-
cation of the function, the entry-point instrumentation will increase constraintFlag to zero, and
the return-point instrumentation will decrease constraintFlag back to -1. Therefore, con-
straintFlag is zero when there is a known active invocation of the function. If this function is
nonrecursive, our instrumentation has failed in every case.

To correct the situation, here is a better MDL implementation. constraintFlag is nonzero if
and only if it has been incremented at entry to the function. Using this knowledge, the new imple-
mentation executes the return-point instrumentation only if the entry-point instrumentation has
been executed. Another way to look at it is that we simply do not decrement constraintFlag
below zero. Of course, if instrumentation occurred while the program was executing within the
function, constraintFlag may still have the incorrect value of zero for the duration of the func-
tion’s current invocation.

  // Being careful in return-point instrumentation

  counter constraintFlag;

  append preInsn func.entry  (* constraintFlag += 1; *)
append preInsn func.return (* if (constraintFlag != 0) constraintFlag -= 1;*)

4.2 Entry-point execution does NOT imply return-point execution

Some functions will return abnormally. For example, suppose there was an exception. Some lan-
guages with exception handling will allow a function to unwind the callstack to an arbitrary depth
until it finds a function that can handle the exception. Another example is the longjmp() library

function from the C programming language. Calling this function essentially unwinds the call-
stack. Yet another example is a case where we did not detect an instruction sequence as a func-
tion-return sequence. There are many ways to return from a function. With optimizing compilers
and creative assembly programmers, we do not claim to know all the instruction sequences that
may be used to return from a function. As a result, we may not have instrumented all return
points of a function.

Let us look at a timer example. Here are the most obvious MDL statements to measure the
execution time of a function. Timer T is on if and only if there is a known active invocation of the
function.

  append preInsn func.entry  (* startProcessTimer(T); *)
  append preInsn func.return (* stopProcessTimer(T);  *)

Let us take a simple walk-through of the above code. On entry to the function, timer T is
started. On return from the function, timer T is stopped. If the program was executing within the
function at the moment of instrumentation, then we will simply stop a timer that was not started,
which has no effect.  Things look good so far.

However, suppose there was a normal entry to the function (timer T started) followed by a
rarely used exceptional return (timer T not stopped). Remember that timers are stopped only if all
startTimer() calls are matched by stopTimer() calls. In this case, the startProcess-

Timer(T) at entry was not matched by the stopProcessTimer(T) at return, so timer T is still
running. On the next invocation of the function, we execute a new startProcessTimer(T), so
we now have two unmatched startProcessTimer(T). At normal return, we execute stopPro-

cessTimer(T) to match the new startProcessTimer(T), but the other unmatched startPro-

cessTimer(T) is still unmatched. So at return, the timer T is still running. In fact, timer T will
always remain running, regardless of whether or not the function is actually being executed.

To solve this problem, we write a better MDL implementation that has a “self-healing” prop-
erty. The next normal execution of the function will correct the instrumentation problems of the

previous exceptional execution of the function1. We accomplish this by making sure that there is
never more than one unmatched startProcessTimer(T). The new MDL code follows. (Unfor-
tunately, there is no simple way to extend this technique to recursive functions.)

  // A self-healing timer metric

  counter T_is_running;  // Flag is 1 if timer T is running

  append preInsn func.entry
    (* if (T_is_running == 0) startProcessTimer(T);

T_is_running = 1;
     *)

  append preInsn func.return
    (* T_is_running = 0;
       stopProcessTimer(T);

1. Note, however, that this “healing” of the instrumentation problem such that subsequent execution will be
correct, doesn’t correct or compensate for spurious accounting which takes place in the interrim.

     *)

We are using the counter T_is_running as a flag that is true when there is an unmatched
startProcessTimer(T). T_is_running is correctly initialized to zero since the timer T is not
running at the moment of instrumentation. At entry to the function, we turn the timer on if and
only if it has not already been turned on. Then we set T_is_running to 1 to indicate that at this
point we are sure the timer is running. Notice that no matter how many times you repeatedly exe-
cute this entry-point instrumentation, there is at most one startProcessTimer(T) that actually
gets executed. At normal return from the function, the stopProcessTimer(T) will match the sin-
gle unmatched startProcessTimer(T). We also set T_is_running to zero since we are sure
there will be no unmatched startProcessTimer(T).

Let us look at the case where we enter the function normally, execute startProcess-

Timer(T), and return exceptionally without executing the return-point instrumentation. Now the
timer T is incorrectly running, and the T_is_running flag is still set to 1. On the next invocation
of the function, the startProcessTimer(T) is not executed. If the function returns normally this
time, the stopProcessTimer(T) will correctly turn off the timer, and the T_is_running flag will
be correctly set to 0.  Our instrumentation is now “healed”.

4.3 Avoid using entry-point post-instrumentation

The sequence in which Paradyn executes instructions at an instrumentation point is as follows.
First, any pre-instrumentation is executed, then the program instructions located at the instrumen-
tation point, and afterwards, any post-instrumentation is executed.  This is illustrated below.

You should try to avoid using entry-point post-instrumentation because the program instruc-
tions found at the point may be branch instructions. These instructions will divert execution flow
away from the post-instrumentation, as illustrated in the figure below. When this happens, any
post-instrumentation will not be executed. In practice, we have not found it necessary to use
entry-point post-instrumentation.

program instructions at point

post1 post2

pre1 pre2

Therefore, you should avoid using any MDL statements that look like the following. Future
versions of Paradyn may even treat them as syntax errors.

  append postInsn func.entry  ...   // RISKY!
  prepend postInsn func.entry  ...   // Just as RISKY!

4.4 Do NOT use return-point post-instrumentation

Return-point post-instrumentation does not work and does not make sense. Therefore, Paradyn
treats any attempt to use such instrumentation as a syntax error.

  append postInsn func.return  ...   // Syntax ERROR!
  prepend postInsn func.return  ...   // Syntax ERROR!

branch instructions

post1 post2

pre1 pre2

diverted execution flow

postInsn ignored

5 CALLSITE INSTRUMENTATION

Callsite instrumentation is commonly used in exclusive timer metrics. These metrics measure the
execution time of a function, but they must exclude the time spent in functions calledby the func-
tion being timed.  Here is a typical example of MDL code for an exclusive timer.

  counter T_is_running;  // Flag is 1 if timer T is running

  append preInsn func.entry
    (* if (T_is_running == 0) startProcessTimer(T);
       T_is_running = 1;
     *)

  append preInsn func.return
    (* T_is_running = 0;
       stopProcessTimer(T);
     *)

  foreach callsite in func.calls {
    append preInsn  callsite
      (* T_is_running = 0;
         stopProcessTimer(T);
       *)

    append postInsn callsite
      (* if (T_is_running == 0) startProcessTimer(T);
         T_is_running = 1;
       *)
  }

Let us take a simple walk-through. At entry to the function, the timer is started. At return
from the function, the timer is stopped. This times the execution of the function. Now we pro-
ceed to exclude the execution times of functions called by this function. We accomplish this by
instrumenting the callsites. At each callsite, we stop the timer before entering the called function,
and we restart the timer after returning from the called function.

Notice in the MDL code given above that the callsite pre-instrumentation is the same as the
function return-point instrumentation and that the callsite post-instrumentation is the same as the
function entry-point instrumentation.  In general, this is true for all exclusive metrics.

5.1 In exclusive metrics, preInsn callsite is analogous to func.return

At a callsite, the function being measured is going to temporarily stop executing until the called
function returns. At function return, the function is going to temporarily stop executing until the
next invocation of the function. To an exclusive metric, the difference in why the function is stop-
ping execution is irrelevant. All that matters is that the exclusive metric must halt until the func-
tion resumes execution.

5.2 In exclusive metrics, postInsn callsite is analogous to func.entry

Immediately following a callsite, the function being measured is going to resume execution after
temporarily stopping to let another function execute. At function entry, the function is going to
resume execution after being temporarily stopped between invocations. To an exclusive metric,
the difference in why the function is resuming execution is irrelevant. All that matters is that the
exclusive metric must be restarted until the function stops execution again.

6 MISCELLANEOUS

This section contains MDL programming tips that could not nicely fit within any of the previous
sections and that could not warrant a separate section by themselves.

6.1 All EventCounter-style metric values must increase monotonically

The values of your EventCounter metrics should not be allowed to decrease in value, no matter

how slightly or for how short an amount of time.2 Paradyn rigorously checks this condition and
will immediately abort if the check ever fails. A typical example of such a metric is one that accu-
mulates a count of the number of bytes transferred by the I/O functions. Sample MDL code is
given below.  The I/O functions give the number of bytes transferred as the return value.

  // Blindly adding function return values

  counter in_sampling;     // Flag is 0 if DYNINSTalarmExpire() is not
                           //  currently being executed

  foreach func in io_funcs {
    prepend preInsn func.return constrained
      (* if (in_sampling == 0) io_bytes += $return; *)
  }

  foreach func in DYNINSTalarmExpire {
    prepend preInsn func.entry
      (* in_sampling = 1; *)

    append preInsn func.return
      (* in_sampling = 0; *)
  }

However, the I/O functions also return -1 if they encounter errors. In fact, it is a general prac-
tice in C programming for functions to return negative values to indicate errors. Therefore, when-
ever an I/O function returns due to error, our instrumentation will add a negativenumber to the

2. SampledFunction-style metrics, of which an example is provided in Section 6.2, allow reported values to
change arbitrarily, however, they must still remain non-negative.

metric value counter io_bytes. Paradyn will detect this decrease in the metric’s value, and
immediately abort, ending your application’s performance measurement session with it.

To solve this problem we simply check the numbers before adding them to their respective
metric value counters.  The MDL code from above is corrected below by adding such a check.

  // Careful when adding function return values

  counter in_sampling;     // Flag is 0 if DYNINSTalarmExpire() is not
                           //  currently being executed

  foreach func in io_funcs {
    prepend preInsn func.return constrained
      (* if ((in_sampling == 0) && ($return > 0)) io_bytes += $return; *)
  }

  foreach func in DYNINSTalarmExpire {
    prepend preInsn func.entry
      (* in_sampling = 1; *)

    append preInsn func.return
      (* in_sampling = 0; *)
  }

6.2 Use SampledFunction-style metrics to track values in a user application

One use of MDL likely to be of particular interest to application or library developers, is the abil-
ity to query a program value and track its evolution. This is generally achieved in two steps: spec-
ification of a query function in the user program which returns the value of interest (similar to
those common for returning the values of private members of C++ classes), and a specification of
when this value should be queried when the metric is enabled.

Because Paradyn uses a shared-memory sampling approach for efficiently extracting perfor-
mance data from the application, a suitable user function needs to be identified as the point when
values should be sampled: this may well be the function (or list of functions) which actually
update the value of interest.

The following example functions from a user program query and update a program value
(which in this case is a global variable, but could have been accessed by any other means):

unsigned int value;
unsigned int program_value() { return (value); }
void update_program_value (unsigned int new_value) { value=new_value; }

The metric itself should be specified of style SampledFunction (and unitsType sampled),
such that the current value is available, rather than the delta from the previously sampled value.
This also has the advantage of allowing values to both increase and decrease, rather than restrict-
ing the value to increase monotonically as is the case with the EventCounter style of metric. Note,
however, that sampled values need to be unsigned integers.

// sample/report a program value after it is updated in a user program

resourceList update_function_list is procedure {
        items { “update_program_value” };        // user’s update function
        flavor { unix };
        library false;
}

metric programValue {
name “program_value”;
style SampledFunction;
units value_units;
unitsType sampled;
aggregateOperator sum;
flavor { unix };

  base is counter {
    foreach func in update_function_list {
      append preInsn func.return constrained
        (* programValue = program_value(); *)     // user’s sampling function
    }
  }
}

Instead of invoking the user function progam_value() to obtain the value of interest, MDL
provides the specially-defined readSymbol  query function to directly read a global variable:

(* programValue = readSymbol(“value”); *) // read global symbol value

Finally, it is worth noting that while the requested value is updated and stored as specified in
the application program’s space, sampling of this value and its reporting by the Paradyn daemon
to the Paradyn front-end happens completely asynchronously. A consequence of this will be that
Paradyn will generally miss data value updates (or repeatedly re-sample the same value) leading
to spurious accumulations (in totals or averages) when compared to sample accumulations by the
user program itself. At best, the reported samples are an approximation of the actual sample
updates requested.

6.3 Instrumenting recursive functions

Recursive functions present particular difficulties for instrumentation metrics, and a number
of the techniques presented in this guide do not readily apply to this class of funtions, or become
considerably more complicated when they must robustly deal with cases of already executing
recursive functions and exceptional returns.

Particular care (and experimentation) is required writing metrics for these cases.

7 CONCLUSION

This paper was written to help you become more proficient in programming with Paradyn’s Met-
ric Description Language (MDL). Specifically, it was designed to save you time and frustrations
in using this new and unusual language. We discussed the two basic MDL metric value types:

counters and timers. We covered tips on how to properly update them in instrumentation. We also
covered how to keep metric values from decreasing.

We have tried to keep this document brief and readable so that you may be encouraged to read
it in its entirety. For more MDL examples, the best source is the paradyn.rc configuration file
provided to you. However, some metrics in the paradyn.rc were implemented using in-depth
knowledge about the functions and/or computing platforms involved, so some of the program-
ming tips developed in this paper were skipped without affecting the correctness of those metrics.
Please keep in mind that such discrepancies do not invalidate the advice given in this paper since
you may not have (or want to acquire) such detailed information about your computing platform.

To simplify the presentation, we provided only MDL code fragments, and not complete metric
implementations. In addition, though constraints are key to instrumentation, none of the MDL
examples given were constrained. All examples were written as if for whole-program metrics.
Again, this was done to simplify the text. The metrics you write, however, will usually need to be
constrained.

Thank you for your interest in Paradyn. We hope this paper was useful to you and that each
future reading continues to provide you new insights. If you have additional questions, please
send email to paradyn@cs.wisc.edu.

■

